
Abstract

There is a relatively large number of papers dealing with
complexity and proof theory issues of infinitely-valued log-
ics. Nevertheless, little attention has been paid so far to
the development of efficient solvers for such logics. In this
paper we show how the technology of Satisfiability Mod-
ulo Theories (SMT) can be used to build efficient auto-
mated theorem provers for relevant infinitely-valued logics,
including Łukasiewicz, G̈odel and Product logics. More-
over, we define a test suite for those logics, and report on
an experimental investigation that evaluates the practical
complexity of Łukasiewicz and Gödel logics, and provides
empirical evidence of the good performance of SMT tech-
nology for automated theorem proving on infinitely-valued
logics.

1 Introduction

There is a relatively large number of papers dealing with
complexity and proof theory issues of infinitely-valued log-
ics (see for example [1, 6, 7, 9] and the references therein).
Nevertheless, little attention has been paid so far to the
development of efficient solvers for such logics. To the
best of our knowledge, most efforts have been devoted to
Łukasiewicz infinitely-valued logic [4, 11]. It is also worth
mentioning the translation from tableaux into mixed inte-
ger programming for certain infinitely-valued logics pro-
posed in [8]. All such efforts are very limited compared
with the work done for Boolean propositional logic. Nowa-
days, Boolean satisfiability solving (SAT) and its extensions
(PBO, QBF, SMT, MaxSAT, . . . ) are recognized and ac-
tive research lines with practical applications in relevant do-

∗Research partially supported by the Generalitat de Catalunya un-
der grant AGAUR 2009-SGR-1434, theMinisterio de Economı́a y Com-
petividadresearch projects AT CONSOLIDER CSD2007-0022, INGENIO
2010, TIN2008-04547, TIN2009-14704-C03-01, and TIN2010-20967-
C04-01/03, and Newmatica INNPACTO IPT-2011-1496-310000.

mains such as software and hardware verification, bioinfor-
matics and planning.

In finitely-valued logics, signed CNF formulas provide a
suitable and effective approach to satisfiability solving.Re-
mind that any finitely-valued formula is polynomially re-
ducible to a satisfiability preserving signed CNF formula,
which can be tested with either a many-valued SAT solver
or a Boolean SAT solver [3]. Unfortunately, there does
not exist a clausal form transformation for dealing with any
infinitely-valued logic. Therefore, it is not clear how SAT
solvers can be used in the infinitely-valued case.

In this paper we propose to use Satisfiability Modulo
Theories (SMT) to solve the satisfiability of infinitely-
valued logics, taking Łukasiewicz, Gödel and Product log-
ics as a case study. As we will see, SMT provides natu-
ral and compact encodings of infinitely-valued logics, and
avoids the need of having a clausal form transformation. On
the other hand, the mature state of SMT technology leads to
very efficient theorem provers, and it would be very costly
to duplicate the same efforts for developing efficient theo-
rem provers based on a tableaux calculus [11] or a sequent
calculus [4].

We also report on an empirical investigation to evaluate
the performance of our approach, and compare the practi-
cal complexity of Łukasiewicz and Gödel logics. To this
end, we have created a test suite which is useful for any t-
norm logic. It consists of a number of theorems of Basic
Logic [9], with tunable hardness, which are easy to gener-
ate.

This paper is structured as follows. Section 2 defines the
notion of many-valued propositional logic, and introduces
Łukasiewicz, Gödel and Product logics. Section 3 presents
a brief introduction to SMT. Section 4 describes SMT-based
theorem provers for Łukasiewicz and Product logics. Sec-
tion 5 presents the test suite we have implemented for t-
norm logics. Section 6 reports on the empirical investiga-
tion. Section 7 presents the conclusions and future work.



2 Infinitely-Valued Logics

We define basic concepts of many-valued propositional
logics, and then introduce the syntax and semantics of
Łukasiewicz, Gödel and Product logics.

Definition 1. A propositional language is a pair
L =< Θ, α >, where Θ is a set of logical connec-
tives andα : Θ → N defines the arity of each connective.
Connectives with arity 0 are called logical constants.
If Θ = {θ1, . . . , θr}, in the following we will de-
note the propositional languageL =< Θ, α > with
< θ1/α(θ1), . . . , θr/α(θr) >.

Given a propositional signatureΣ, the setLΣ of L-
formulas overΣ is inductively defined as the smallest set
with the following properties: (i)Σ ⊆ LΣ; (ii) if θ ∈ Θ and
α(θ) = 0, thenθ ∈ LΣ; and (iii) if ϕ, . . . , φm ∈ LΣ, θ ∈ Θ
andα(θ) = m, thenθ(ϕ, . . . , φm) ∈ LΣ.

Definition 2. If L =< Θ, α > is a propositional language,
N is a truth value set andA assigns to eachθ ∈ Θ a function
Aθ : Nα(θ) → N , then we call a pairA =< N,A > a
matrix forL.

A pair L =< L,A > consisting of a propositional lan-
guage and a matrix is called a many-valued logic.

Many-valued logics are equipped with a non-empty sub-
setD of N called the designated truth values which are the
truth values that are considered to affirm satisfiability.

Definition 3. Let L be a many-valued logic. An interpre-
tation is a functionI : Σ → N . I is extended to arbitrary
formulasφ in the usual way:

1. If φ is a logical constant, theI(φ) = A(φ).

2. If φ = θ(ϕ1, . . . , φr) , then I(θ(ϕ1, . . . , φr)) =
Aθ(I(ϕ1), . . . , I(φr)).

A formula φ is satisfiable iff there is an interpretation
such thatI(φ) ∈ D, and it is a tautology iffI(φ) ∈ D for
every interpretation.

In the rest of the section we concentrate on t-norm based
fuzzy propositional logics [9]. More specifically, we fo-
cus on Łukasiewicz logic (Ł), Gödel logic (G) and Prod-
uct logic (Π), which are the most representative fuzzy log-
ics and can be obtained from Basic Fuzzy Logic (BL) by
adding one or two additional axioms. Moreover, each con-
tinuous t-norm is an ordinal sum of isomorphic copies of
the t-norms of Ł,G andΠ [10].

From now on, we assume that a formulaφ of an
infinitely-valued logic is satisfiable iff there is an interpre-
tation such thatI(φ) = 1, and it is a tautology iffI(φ) = 1
for every interpretation.

Definition 4. The language of Łukasiewicz logic is given
by

LŁuk
=< ¬/1,→ /2,∧/2,∨/2,⊙/2,⊕/2> .

Subsets of the set of connectives such as{→,¬} and
{⊕,¬} are functionally complete. Nevertheless, we con-
sider the whole set of connectives to illustrate how they are
encoded into SMT. We refer to¬ as negation, refer to→
as implication, refer to∧ as weak conjunction, refer to∨
as weak disjunction, refer to⊙ as strong conjunction, and
refer to⊕ as strong disjunction.

Definition 5. The Łukasiewicz infinitely-valued logic is the
many-valued logic such thatN is the real unit interval[0, 1],
L = LŁuk

,D = {1}, and the matrix is given by:

A¬(x) = 1− x
A→(x, y) = min{1, 1− x+ y}
A∧(x, y) = min{x, y}
A∨(x, y) = max{x, y}
A⊙(x, y) = max{0, x+ y − 1}
A⊕(x, y) = min{1, x+ y}

Definition 6. The language of Gödel logic is given by

LG =< ¬/1,→ /2,∧/2,∨/2 > .

Definition 7. The Gödel infinitely-valued logic is the
many-valued logic such thatN is the real unit interval[0, 1],
L = LG,D = {1}, and the matrix is given by:

A¬(x) =

{

1 if x = 0
0 if x > 0

A→(x, y) =

{

1 if x ≤ y
y if x > y

A∧(x, y) = min{x, y}
A∨(x, y) = max{x, y}

In Gödel logic we do not make distinction between weak
and strong conjunction because both are interpreted by the
same truth functions.

Definition 8. The language of Product logic is given by

LΠ =< ¬/1,→ /2,∧/2,∨/2,⊙/2,⊕/2> .

Definition 9. The Product infinitely-valued logic is the
many-valued logic such thatN is the real unit interval[0, 1],
L = LΠ,D = {1}, and the matrix is given by:

A¬(x) =

{

1 if x = 0
0 if x > 0

A→(x, y) =

{

1 if x ≤ y
y/x if x > y

A∧(x, y) = min{x, y}
A∨(x, y) = max{x, y}
A⊙(x, y) = x · y
A⊕(x, y) = x+ y − x · y



3 Satisfiability Modulo Theories (SMT)

An SMT instance is a generalization of a Boolean for-
mula in which some propositional variables have been re-
placed by predicates with predefined interpretations from
background theories. For example, a formula can contain
clauses like, e.g.,p ∨ q ∨ (x + 2 ≤ y) ∨ (x > y + z),
wherep andq are Boolean variables, andx, y andz are real
variables. Predicates over non-Boolean variables, such as
linear real inequalities, are evaluated according to the rules
of a background theory. Examples of theories include linear
real or integer arithmetic, arrays, bit vectors, etc., or combi-
nations of them.

Formally speaking, atheoryis a set of first-order formu-
las closed under logical consequence. The SMT problem
for a theoryT is: given a first-order formulaF , determine
whether there is a model ofT ∪ {F}.

Although an SMT instance can be solved by encoding it
into an equisatisfiable SAT instance and feeding it to a SAT
solver, currently most successful SMT solvers are based
on the integration of a SAT solver and aT -solver, that is,
a decision procedure for the given theoryT . In this so-
called lazy approach, while the SAT solver is in charge of
the Boolean component of reasoning, theT -solver deals
with sets of atomic constraints inT . The main idea is
that theT -solver analyzes the partial model that the SAT
solver is building, and warns it about conflicts with theory
T (T -inconsistency). This way, we are hopefully getting
the best of both worlds: in particular, the efficiency of the
SAT solver for the Boolean reasoning and the efficiency of
special-purpose algorithms inside theT -solver for the the-
ory reasoning. See [13] for a survey on this approach.

Among the theories considered in the SMT library we
are interested in the following logics [2]:

• QF LRA: quantifier-freelinear real arithmetic. In
essence, closed quantifier-free formulas with Boolean
combinations of inequalities between linear polynomi-
als over real variables.

• QF NRA: quantifier-freenon-linear real arithmetic. It
is like QF LRA logic but allowing non-linear expres-
sions.

4 Automated Theorem Provers for Infinitely-
Valued Logic

The aim of this section is to show that SMT technol-
ogy can indeed be used to build automated theorem provers
for infinitely-valued logics, taking Łukasiewicz and Product
infinitely-valued logics as a case study. As we will see, such
logics can be encoded into SMT in a natural and compact
way. We do not show Gödel logic due to the lack of space,
but its formulation is similar.

4.1 The SMT-based Theorem Prover for
Łukasiewicz Infinitely-Valued Logic

We describe a theorem prover capable of determining
whether a formulaφ of the Łukasiewicz infinitely-valued
logic is a tautology. To this end, we develop a satisfiability
checker for Łukasiewicz logic, and we ask whether there is
an interpretationI such thatI(φ) < 1. If such an interpre-
tation does not exist, thenφ is a tautology.

Figure 1 shows the code of the SMT-based theorem
prover for Łukasiewicz logic in the SMT-LIB language v2.0
under QFLRA [2]. We find three sections. The first one
determines the logic to be used: quantified free linear real
arithmetic (QFLRA). The second one declares the connec-
tives of Łukasiewicz logic, and the variables occurring in
the formula. Notice that the variable domains are restricted
to the real unit interval. Theite (if-then-else) SMT func-
tion(ite Bool s s) returns its second argument or its
third depending on whether its first argument is true or not.

The third one contains the following query: Is there an
interpretationI of the formulaφ = (x ⊕ x) ∨ (y ⊕ y) →
(x ∨ y) ⊕ (x ∨ y) such thatI(φ) < 1. φ is a tautology be-
cause the theorem prover returns unsatisfiable. If we would
like to determine whether another formula is a tautology,
we should only replace the formula and leave the rest of the
program unchanged.

4.2 The SMT-based Theorem Prover for Product
Infinitely-Valued Logic

Figure 2 shows the code of the SMT-based theorem
prover for Product logic in the SMT-LIB language v2.0 un-
der QFNRA [2]. We also find three sections. The first
one determines the logic to be used: quantified free non-
linear real arithmetic (QFNRA). The second one declares
the connectives. Observe that now the defined functions
correspond to the truth functions of Product logic. The third
one contains the query, which corresponds to axiom A2 of
BL for n = 1, defined in the next section.

5 A test Suite for t-Norm Logics

Basic Logic (BL) offers a framework for defining the
most representative fuzzy logics by adding additional ax-
ioms to the axioms ofBL. This means that the axioms of
BL defined in [9] are valid in all the t-norm based fuzzy
logics, including Ł,G andΠ. Our test suite is formed by
instances of the first six axioms ofBL defined as follows:

• (A1) (ϕn → ψn) → ((ψn → χn) → (ϕn → χn))

• (A2) (ϕn ⊙ ψn) → ϕn

• (A3) (ϕn ⊙ ψn) → (ψn ⊙ ϕn)



(set-logic QF_LRA)

; min(x,y)
(define-fun min ((x Real) (y Real)) Real
(ite (> x y) y x))
; max(x,y)
(define-fun max ((x Real) (y Real)) Real
(ite (> x y) x y))
; strong disjunction: sdis(x,y) = min{1,x+y}
(define-fun sdis ((x Real) (y Real)) Real
(min 1 (+ x y)))
; strong conjuction: scon(x,y) = max{0,x+y-1}
(define-fun scon ((x Real) (y Real)) Real
(max 0 (- (+ x y) 1)))
; weak disjunction: wdis(x,y) = max{x,y}
(define-fun wdis ((x Real) (y Real)) Real
(max x y))
; weak conjunction: wcon(x,y) = min{x,y}
(define-fun wcon ((x Real) (y Real)) Real
(min y x))
; negation: neg(x) = 1 - x
(define-fun neg ((x Real)) Real
(- 1 x))
; implication: impl(x,y) = min{1,1-x+y}
(define-fun impl ((x Real) (y Real)) Real
(min 1 (- (+ 1 y) x)))
(declare-fun x () Real)
(declare-fun y () Real)
(assert (>= x 0))
(assert (<= x 1))
(assert (>= y 0))
(assert (<= y 1))

(assert (< (impl (wdis (scon x x) (scon y y))
(scon (wdis x y) (wdis x y))) 1))

(check-sat)

Figure 1. Code of the SMT-based theorem
prover for Łukasiewicz infinitely-valued logic.

(set-logic QF_NRA)

; min(x,y)
(define-fun min ((x Real) (y Real)) Real
(ite (> x y) y x))

; max(x,y)
(define-fun max ((x Real) (y Real)) Real
(ite (> x y) x y))

; strong disjunction: sdis(x,y) = x+y - x*y
(define-fun sdis ((x Real) (y Real)) Real
(- (+ x y) (* x y)))

; strong conjuction: scon(x,y) = x*y
(define-fun scon ((x Real) (y Real)) Real
(* x y))

; weak disjunction: wdis(x,y) = max{x,y}
(define-fun wdis ((x Real) (y Real)) Real
(max x y))

; weak conjunction: wcon(x,y) = min{x,y}
(define-fun wcon ((x Real) (y Real)) Real
(min y x))

; negation: neg(x) = 1 if x=0; 0 if x>0
; we assume x in [0,1]
(define-fun neg ((x Real)) Real
(ite (= x 0) 1 0))

; implication: impl(x,y) = y/x if x>y; 1 otherwise
(define-fun impl ((x Real) (y Real)) Real
(ite (> x y) (/ y x) 1))

(declare-fun x () Real)
(declare-fun y () Real)
(declare-fun x1 () Real)
(assert (>= x 0) )
(assert (<= x 1) )
(assert (>= y 0) )
(assert (<= y 1) )

(assert (< (impl (scon x y) x) 1))
(check-sat)

Figure 2. Code of the SMT-based theorem
prover for Product infinitely-valued logic.

Figure 3. Łukasiewicz logic: Axiom A2

Figure 4. Łukasiewicz logic: Axiom A5a

• (A4) (ϕn ⊙ (ϕn → ψn)) → (ψn ⊙ (ψn → ϕn))

• (A5a) (ϕn → (ψn → χn)) → ((ϕn ⊙ ψn) → χn)

• (A5b) ((ϕn ⊙ ψn) → χn) → (ϕn → (ψn → χn))

• (A6) ((ϕn → ψn) → χn) → (((ψn → ϕn) →
χn) → χn)

whereφ1 = p andφn = p ⊙ φn−1
i

for φ ∈ {ϕ, ψ, χ}; and
p is a propositional variable.

6 Empirical Investigation

We performed an empirical evaluation of our SMT-based
theorem provers. We ran experiments on a machine with 2.5
GHz and 0.5G mem. The SMT solver used in our experi-
mentation is Z3 [5] (version 3.2) with the QFLRA logic.
We do not show results for Product Logic because the im-
plementation of QFNRA in Z3 is limited1.

Figure 3 shows the results for axiom A2 ofBL, rang-
ing from n = 1 to n = 300 with increments of 10, with
the described SMT-based theorem prover for Łukasiewicz
infinitely-valued logic. Figure 4 is like Figure 3 but for ax-
iom A5a ofBL, and Figure 5 is like Figure 3 but for axiom
A6 of BL.

1In fact, Z3 does not support expressions likex/y where bothx and
y are variables. We have made our experiments by indirectly encoding
division with multiplication and auxiliary variables, butstill Z3 has just
been able to solve a few small instances.

Figure 5. Łukasiewicz logic: Axiom A6

 0

 0.2

 0.4

 0.6

 0.8

 1

 1.2

 1.4

 0  50  100  150  200  250  300

ru
n 

tim
e 

(s
ec

on
ds

)

n

Theorem 2

z3

Figure 6. Product logic: Axiom A2

Figure 6 shows the results for axiom A2 ofBL, ranging
from n = 1 to n = 300 with increments of 10, with the
described SMT-based theorem prover for Gödel infinitely-
valued logic. Figure 7 is like Figure 6 but for axiom A5a of
BL, and Figure 8 is like Figure 6 but for axiom A6 ofBL.

In all the figures, thex-axis represents the value ofn and
they-axis represents the time, in seconds, needed to prove
the theorem. Due to the lack of space, we do not display
results for the other axioms of BL, but we have obtained
similar results.

Observe that the difficulty of the instances increases with
n for both Łukasiewicz and Gödel logics. So, we can ad-
just the hardness of the benchmarks. This is a desired prop-
erty for comparing the scaling behavior of different theorem
provers. Also notice that the benchmark theorems hold for
any fuzzy logic, not only for Łukasiewicz and Gödel logics.

On the other hand, observe that Łukasiewicz theorems
are harder to prove in practice, despite that the tautology
problem is NP-complete in both logics. This is due to the
overhead of the arithmetic operations in the truth functions
of Łukasiewicz logic. We claim that the results for Product
logic will be even harder because the overhead of non-linear
operations is bigger.

Notice that there are some peaks in the plots. The in-
crease is not completely monotone because of the heuristics
implemented in SMT solvers.

It is worth commenting that the theorems are proved in a
reasonable amount of time taking into account the length
of the instances. This performance profile is difficult to

Figure 7. Product logic: Axiom A5a

Figure 8. Product logic: Axiom A6

reach using a Prolog implementation as it was done for
Łukasiewicz logic in [12].

7 Conclusions

We have presented how SMT technology can be used to
define theorem provers for fuzzy logics, built a test suite for
those logics, and performed a preliminary empirical eval-
uation. We would like to highlight that the code of the
described theorem provers provides a natural and compact
representation of the logics. Moreover, the code can be un-
derstood without needing a deep knowledge on SMT, and
the described theorem provers can easily be adapted to other
fuzzy logics.

Our work can be seen as a position paper in the sense
that opens a new research line, consisting in exploiting SMT
technology in fuzzy logics. To the best of our knowledge
there has not been established before a link between SMT
and many-valued logics.

There are a number of tasks and open questions that we
propose as future work:

• Perform an empirical comparison between described
theorem provers for infinitely-valued logics and our
SMT-based approach. In particular, it would be inter-
esting to compare with the approach described in [8].

• Perform experiments for Product logic using an SMT-
based theorem prover with an efficient implementation
of non-linear real arithmetic.

• Create a webpage containing a repository with bench-
marks and theorem provers for fuzzy logics.

• Solve real applications with SMT-based theorem
provers. The non-existence of fast and modern theo-
rems provers has limited so far the potential of fuzzy
logics in some real-world applications.

• Stimulate the development of efficient theorem provers
and the creation of challenging benchmarks by orga-
nizing a competition.

• Evaluate the performance profile of SMT technology
for finitely-valued logics.

